Computer Science Principles

Creativity and Processing

**Goal:** The purpose of this exercise is to give you a chance to be creative with a partner in Processing.

Below are four examples; notice that these are short programs that do something worth watching for a moment.

**Examples**

Below are four Processing programs like the kind expected in this assignment. You can use anything from these programs, but keep in mind that small changes from one of these programs is NOT creative ... it’s derivative! Grading is based on creativity, good use of the language and your **comments**.

**Random Lines**

Here is a program that draws batches of 10 random lines of either white or red, chosen at random.

```java
/* Written by ms. evans
   Last edited: 03/01/2015
   Displays groups of 10 random lines
   in either red or white (randomly)
 */

// sets up the 500x500 window, sets the
// background black, slows down the frame
// rate and sets the weight of the lines to 2
void setup(){
    size(500, 500);
    background(0);
    frameRate(5);
    strokeWeight(2);
}

void draw(){
    // 50% of the time it will draw
    // white lines, 50% of the time
    // it will draw red lines
    if(random(0,1) < 0.5){
        stroke(255);
    } else {
        stroke(255, 0, 0);
    }

    // draws 10 lines in random positions
    for(int i=0; i<10; i++){
        line(random(1,499), random(1,499), random(1, 499), random(1, 499));
    }
}
```
Seattle Rain
This program simulates the Seattle rain. This program generates droplets of a random size and one of two different (but very similar) random colors. The slow frame rate makes the action develop in kind of a hypnotic way, like Seattle rain.

/* Written by ms. evans
   Last edited: 03/01/2015
   Simulates the Seattle rain with
   slow appearing ellipses
*/

// sets up the 500x500 window, sets the
// background black, slows down the frame
// rate and sets the fill to black.
void setup(){
  size(500, 500);
  background(0);
  frameRate(5);
  fill(0);
}

void draw(){
  // 50% of the time it will draw
  // a darker line, 50% of the time
  // it will draw a lighter line
  if(random(0,1) < 0.5){
    stroke(180, 180, 255);
  }
  else {
    stroke(200, 200, 255);
  }

  // draws 10 circles at a time in a random spot
  // and with a random size; the size variable is used
  // to make it a circle instead of an ellipse
  for(int i=0; i<10; i++){
    float size = random(10, 20);
    ellipse(random(1,499), random(1,499), size, size);
  }
}
Programming A Heart Beat

Here a heart has been drawn with a few basic shapes, and colored red. The program then turns down the red some, and then turns it back up again, to simulate a heart beat.

/* Written by ms. evans
   Last edited: 03/01/2015
   Simulates a beating heart */

int dir = -1; // controls whether the heart gets lighter or darker
int tint = 255; // starts the tint off to dark

// sets up the 500x500 window, sets the
// background gray, speeds up the frame rate,
// turns off the stroke and sets the fill to red.
void setup(){
  size(500, 500);
  background(64);
  frameRate(15);
  noStroke();
  fill(0);
}

void draw(){
  // if the tint is low, start increasing it
  if(tint < 180){
    dir = 1;
  }

  // if the tint is high, start decreasing it
  if(tint > 255){
    dir = -1;
  }

  // this increases or decreases the tint
  // based on the value of dir
  tint = tint + dir * 5;

  // draw the heart
  heart(tint);
}

// creates the heart shape
void heart(int tint){
  fill(tint, 0, 0);
  ellipse(280, 250, 100, 100);
  ellipse(380, 250, 100, 100);
  triangle(153, 276, 348, 276, 256, 488);
  rect(248, 250, 20, 20);
}
Walking Man

In this program a stick figure is animated to walk forward ... will he fall into the abyss?? Run it to find out. Notice that the body can be moved by using our standard move-to-the-right techniques, but the legs must work differently.

```cpp
/* Written by ms. evans
   Last edited: 03/01/2015
   A walking stick figure; will he fall in? */

int walk = 0; // controls which leg is moving faster
int x = 0; // controls the x position of the figure
int leftStep = 0; // controls the left leg's position
int rightStep = 0; // controls the right leg's position

// sets up the 500x500 window, speeds up
// the frame rate, smooths out the lines, turns
// off fill and sets the strokeWeight to 3
void setup(){
  size(500, 500);
  frameRate(30);
  smooth();
  noFill();
  strokeWeight(3);
}

void draw(){
  background(0);
  level();

  // chooses which leg is moving faster
  if(walk == 0){
    leftStep = leftStep + 2;
  } else {
    rightStep = rightStep + 2;
  }

  // switches the legs
  if(abs(leftStep-rightStep) >= 15){
    walk = 1 - walk;
  }

  figure(x);
  x = x + 1;
}

// draws the level
void level(){
  stroke(255, 255, 0);
  line(50, 170, 302, 170);
  line(302, 170, 300, 300);
  line(333, 170, 450, 170);
  line(333, 170, 333, 300);
}

// draws the figure at position x
void figure(int x){
  stroke(255, 0, 0);
  ellipse(100+x, 100, 20, 20);
  line(100+x, 110, 100+x, 140);
  line(100+x, 125, 80+x, 110);
  line(100+x, 125, 120+x, 110);

  line(100+x, 140, 90+leftStep, 170);
  line(100+x, 140, 110+rightStep, 170);
}```